

ROSY: An elegant language to teach the pure reactive nature of robot programming

Hugo Pacheco

Universidade do Minho & INESC TEC

Portugal

Email: hpacheco@di.uminho.pt

Nuno Macedo

Universidade do Minho & INESC TEC

Portugal

Email: nfmacedo@di.uminho.pt

Abstract—Robotics is incredibly fun and is long recognized as a great way to teach programming, while drawing inspiring connections to other branches of engineering and science such as maths, physics or electronics. Although this symbiotic relationship between robotics and programming is perceived as largely beneficial, educational approaches often feel the need to hide the underlying complexity of the robotic system, but as a result fail to transmit the reactive essence of robot programming to the roboticists and programmers of the future.

This paper presents **ROSY**, a novel language for teaching novice programmers through robotics. Its functional style is both familiar with a high-school algebra background and a materialization of the inherent reactive nature of robotic programming. Working at a higher-level of abstraction also teaches valuable design principles of decomposition of robotics software into collections of interacting controllers. Despite its simplicity, **ROSY** is completely valid Haskell code compatible with the **ROS** ecosystem.

We make a convincing case for our language by demonstrating how non-trivial applications can be expressed with ease and clarity, exposing its sound functional programming foundations, and developing a web-enabled robot programming environment.

I. INTRODUCTION

Robotics, though a multi-disciplinary area of engineering and science, shares a unique symbiotic relationship with computer science. As robots are increasingly put to solve more complex tasks, the more important it becomes to look carefully into the software that runs inside them, and in particular to the programming languages that bring them to life. At the same time, programming a robot can grant computer science an often missing practical appeal, with a pedagogical potential long recognized by computer science educators [1].

Various studies corroborate that robotics can indeed mitigate the abstract nature of programming [2] and provide a more creative environment for teaching programming concepts focused on hands-on problem solving [3] and for coding rich behavior using basic code structures [4], while still requiring reasoning about relevant computational concepts such as software modularity and communication [5].

Yet, the richness of robot programming is synonym of its real-world complexity, which renders standard robotic frameworks unfeasible for use in a pedagogical setting and has been motivating the proposal of various educational robot programming frameworks. As the synergy between robotics and programming deepens, education is the perfect place to experiment novel approaches that can shape the robot

programming languages of the future [6]. It also carries a timely opportunity to transmit good design practices to new generations of programmers, that are receptive to novel languages as long as these allow to quickly build applications [7].

As a step towards realising this vision, we advocate that languages for teaching robotics to novice programmers shall:

- be compatible with standard robotic practices and seamlessly connect to existing robotic infrastructures;
- adopt a simple declarative programming style and provide a pure cause-and-effect interface emphasizing the essence of what it means to program a reactive system;
- rely on a general-purpose programming language with good tool support, so that advanced programming features can be gradually introduced and acquired programming skills can naturally transfer to other domains.

In Section II, we argue that state-of-the-art languages fail, to some extent, to exhibit these characteristics. This justifies the proposal of **ROSY**, a simple yet powerful reactive programming language, presented in Section III. To ease adoption by novice programmers, **ROSY** is supported by a browser-based development environment, described in Section IV. As an embedded domain-specific language, **ROSY** supports the full power of higher-order functional programming offered by the host Haskell language, and is connected to **ROS**, one of the most popular robotic middlewares. The mechanics behind its implementation is presented in Section V. Section VI concludes the paper and leaves directions for future work.

II. THE PEDAGOGY OF ROBOT PROGRAMMING LANGUAGES

Programming robots is a particularly complex task, where one has to deal not only with the continuous and real-time aspects of the physical world, but also with heterogeneous architectures and complex communication paradigms. To minimize frustration of novice programmers, several pedagogical languages and approaches have been proposed over the years. This section reviews and discusses such related work.

A. The Robot Operating System

Robotic middlewares have been developed to ease the programming of robots, abstracting hardware and communication details and promoting modularity. The Robot Operating System (**ROS**) [8] is one such middleware, possibly the most popular, and defines an architecture through which components,

called *nodes* in ROS, can communicate with each other by publishing to or subscribing from data sources, called *topics*. Other than that, individual nodes are programmed in general-purpose languages, typically C++ or Python. The popularity of ROS is fueled by a very dynamic community and an open-source policy that encourages code re-use, and a large package database ranging from educational to industrial applications.

As a pedagogical example, consider the well-known TurtleBot2 robot, whose controller is programmed in ROS-powered C++. A main node controls its Kobuki mobile base publishing odometry information and data collected from a set of sensors (collision, cliff and wheel drop sensors, plus a few buttons), and subscribing to commands that control its (linear and angular) velocity and the color of a set of LEDs. ROS allows the definition of custom message types, from which source code is automatically generated, and Kobuki defines such message types for the relevant events. Below is a minimal example of a ROS application that subscribes to bumper events and plays an error sound when a bumper is pressed:

Snippet 1 (Play a sound on collision).

```
ros::Publisher pub;

void cb (const kobuki_msgs::BumperEventConstPtr& b) {
  if (b->state==kobuki_msgs::BumperEvent::PRESSED) {
    kobuki_msgs::Sound s;
    s.value = kobuki_msgs::Sound::ERROR;
    pub.publish(s); } }

int main(int argc, char** argv){
  ros::init(argc, argv, "play");
  ros::NodeHandle nh;
  ros::Subscriber sub = nh.subscribe(
    "/mobile_base/events/bumper", 10, cb);
  pub = nh.advertise<kobuki_msgs::Sound>(
    "/mobile_base/commands/sound", 10);
  ros::spin();
  return 0; }
```

Even this minimal snippet is clearly non-trivial for novice programmers, obfuscating the truly reactive nature of the controller. Besides having to get around advanced linguistic features such as pointers, namespaces or templates, to manipulate topics the programmer must first explicitly `subscribe` to bumper events and `advertise` that sound commands will be published, considering the buffer size for incoming and outgoing messages. The programmer must also reason about how topics are processed, by registering a callback on the `ros::Subscriber` that will `publish` an error sound command if a bumper pressed event is read, and when topics are processed, in this simplest case using the ROS `spin` primitive that periodically processes callbacks for the queued messages.

B. Visual robot programming languages

To tame the complexity of programming robots in fully-fledged general-purpose programming languages, a myriad of frameworks aimed at novice robot programming using visual programming languages have been proposed [9]. Many of these follow a block-based approach [10]–[14], where predefined blocks with varied colors and edges can be put together like pieces of a puzzle to define a robotic controller.




Fig. 1: A simple block that draws a square.

Aside from discussions on whether block-based approaches constitute “real” programming or their programming experience transfers to “real” textual languages [15], [16], and with due exceptions such as [14], the vast majority of block-based robot programming approaches adopt an imperative mindset: primitive blocks execute individual predefined tasks, such as moving forward during a certain period of time or for a certain distance; and combining blocks amounts to performing sequences of tasks. For example, the TurtleBot3 can be programmed via a block-based interface in which we can write a block similar to the one from Figure 1. Albeit simple, this example block hides away the reactive nature of the robotic system – all the sensor and command behavior is encapsulated within the black-box primitive blocks.

C. Functional reactive programming languages

Reactive programming [17] is a programming paradigm organised around information producers and consumers, that can naturally bring out the intrinsically reactive nature of cyber-physical systems. A particularly active line of research, known as functional reactive programming (FRP) [18], focuses on streams of information as the central reactive abstraction, and advocates a declarative approach to manipulate streams at a high-level of abstraction supported by the pure equational reasoning of functional languages such as Haskell.

Functional languages, as a form of algebra, are a good fit for introductory programming [7], and many pedagogical FRP approaches have been proposed for programming interactive games and animations [6], [19]–[21]. Much classical work has also proposed FRP for programming robots [22]–[24].

In FRP, time is typically explicit and conceptually continuous; the execution of the system is then carried out by sampling all streams synchronously at the rate of an external and global clock. The ROSHASK [25] Haskell ROS library promotes the modularity and expressiveness of FRP, while remaining faithful to the asynchronous nature of ROS, by adopting a more pragmatic approach centered on manipulating asynchronous topics in their entirety.

D. Towards a pedagogical robot FRP language

Despite the declarative nature of functional programming and the focus on events of reactive programming, the combinatorial FRP style is not beginner friendly, as it tends to swallow entire programs and resorts to advanced higher-order features to separate reactive code (referring to entire topics) from non-reactive code (referring to individual events).

In this paper, we advocate that a toned-down FRP language, focused on individual events, represents a sweet spot of

introductory robot programming. Making time implicit is a big win, as it liberates novice programmers from specific FRP syntax and invites them to simply write pure functions. To retain some of the expressiveness of FRP, these functions have then an intuitive interpretation as operations on data streams.

III. THE ROSY LANGUAGE

The ROSY language presents itself as a natural dialect for bringing robots alive using nothing more than plain mathematics, while promoting good software design practices. In this section, we make a case for how its declarative nature allows creating robotic controllers in an intuitive and painless way, and informally present its defining features through a collection of examples of increasing complexity that illustrate how to control a TurtleBot2 robot. The types and associated fields used in the examples will therefore spell the standard Kobuki ROS message types. The ROSY website¹ offers a modern integrated development environment, including an editor, help guides, and executable versions of all the examples shown in this section and more.

For readers not familiar with Haskell syntax, all the functions and operators not defined in this paper are standard and their definition can be found at <https://hoogle.haskell.org>. The documentation for respectively colored ROSY-specific functions and types is available at the ROSY website.

A. A ROSY primer

In ROSY, we can control a robot by writing pure functions that receive sensor information from the robot and react by sending commands back to the robot. To make a robot move forward at a constant velocity of 0.5 m/s, we can simply write:

Example 1 (Move forward).

```
move :: Velocity
move = Velocity 0.5 0

main = simulate move
```

The `move` function models our controller (where the `Velocity` of the robot is separated into its linear and angular components), and the `main` function is ROSY-specific syntax to `simulate` our controller (that will be elided from now on). Still, the astute reader may fittingly ask “for how long are we telling the robot to move forward?” Being ROSY a reactive programming language, the answer is not “once” or “for a certain period of time”, but actually “forever”. The intuition is that a controller is a function that unceasingly listens for inputs, and for each received input produces an output, what grants ROSY programs an implicit notion of time. In this case, since `move` receives no inputs, it will produce `Velocity` outputs at a fixed rate.

To make things a bit more interesting, imagine that we want the robot to accelerate forward with non-constant velocity. We can achieve this behavior by making sure to increase the robot’s velocity at each point in time:

Example 2 (Accelerate forward).

```
accelerate :: Velocity -> Velocity
accelerate (Velocity v1 va) = Velocity (v1+0.5) va
```

Note that the same `Velocity` type has different input and output meanings. This second `accelerate` controller is a function that repeatedly asks the robot for its current linear velocity, and commands the robot to increase it by 0.5 m/s.

As our robot is moving forward, what if it hits a wall? We can naturally express multiple ROSY controllers that react to distinct events. For instance, we can make the robot play an error sound when one of its bumpers is pressed, what will happen on contact with a wall:

Example 3 (Accelerate and play a sound on collision).

```
play :: Bumper -> Maybe Sound
play (Bumper _ Pressed) = Just ErrorSound
play (Bumper _ Released) = Nothing

accelerateAndPlay = (accelerate,play)
```

In this example, we define a composite `accelerateAndPlay` controller by simply pairing together `accelerate` and `play`. Note that these two functions are not required to execute at the same time: `accelerate` runs on periodic robot information, though `play` waits for `Bumper` events. The two controllers will execute in parallel, effectively combining both behaviors. To be able to play a sound only when a bumper is `Pressed`, and not `Released`, the `play` function may or may not produce a `Sound` command. The `play` ROSY controller displays the same behavior as the ROS one encoded in Snippet 1, but here its reactive nature is clear from the type declaration.

Even though the controller from Example 3 detects when the robot hits a wall, it will continue to push against the wall, likely reaching a deadlock. With controllers as pure functions that react to events as they happen, there is no easy way to make a controller remember some event in the past. This contrasts with traditional imperative robot programming languages, where we could use a global variable to, e.g., memorize when the robot has hit a wall and, from then on, change its behavior. Like other pedagogical functional languages [19], we grant ROSY controllers a notion of `Memory`, that can be used to, e.g., remember the robot’s moving direction:

Example 4 (Accelerate forward and backwards on collision).

```
type Hit = Bool

reverseDir :: Bumper -> Memory Hit
reverseDir _ = Memory True

accelerate :: Memory Hit -> Velocity -> Velocity
accelerate (Memory hit) (Velocity v1 va) = if hit
  then Velocity (v1-0.5) va
  else Velocity (v1+0.5) va

forwardBackward = (reverseDir,accelerate)
```

In the above example, the controller hands over its memory to `accelerate`, that uses it to determine in which direction to move. To reconcile memory with pure functional programming, controllers that wish to change the memory are expected to return it explicitly as an output. The `Hit` boolean memory

¹<https://haskell.lsd.di.uminho.pt/rosy>




Fig. 2: The random walker application in ROSY.

will be false by default, and changed to true by `reverseDir` when a bumper event occurs.

B. Revisiting ROS controllers

For a more complete and realistic example, we now encode the popular Kobuki random walker controller in ROSY, while trying to stay faithful to the C++ ROS implementation²:

- On a bumper or cliff event, the robot blinks one of its LEDs orange and decides to change its direction;
- On a wheel drop event, the robot blinks both LEDs red and decides to stop moving while the wheel is in the air;
- When changing direction, it randomly decides on an angle between 0° and 180° and on a left/right direction. Depending on a fixed angular velocity, it estimates how many seconds it shall turn;
- A sequential loop routinely performs the adequate action depending on the state of the controller. It commands the robot to either: go forward, stop moving, or turn in a given direction for a number of seconds.

Example 5 (Random walker).

```
data Mode = Go | Stop | Turn Double Seconds
data ChgDir = ChgDir -- change direction

vel_lin = 0.5
vel_ang = 0.1

bumper :: Bumper -> (Led1, Maybe ChgDir)
bumper (Bumper _ st) = case st of
  Pressed -> (Led1 Orange, Just ChgDir)
  Released -> (Led1 Black, Nothing)

cliff :: Cliff -> (Led2, Maybe ChgDir)
cliff (Cliff _ st) = case st of
  Hole -> (Led2 Orange, Just ChgDir)
  Floor -> (Led2 Black, Nothing)

wheel :: Wheel -> (Led1, Led2, Memory Mode)
wheel (Wheel _ st) = case st of
  Air -> (Led1 Red, Led2 Red, Memory Stop)
  Ground -> (Led1 Black, Led2 Black, Memory Go)

chgdir :: ChgDir -> StdGen -> Seconds
      -> Memory Mode
chgdir _ r now = Memory (Turn dir time)
  where
    (b, r') = random r
    (ang, _) = randomR (0, pi) r'
    dir = if b then 1 else -1
    time = now + doubleToSeconds (ang/vel_ang)

spin :: Memory Mode -> Seconds
```

```
-> (Velocity, Memory Mode)
spin m@(Memory Stop) _ = (Velocity 0 0, m)
spin m@(Memory (Turn dir t)) now | t > now =
  (Velocity 0 (dir*vel_ang), m)
spin m _ = (Velocity vel_lin 0, Memory Go)

randomWalk = (bumper, cliff, wheel, chgdir, spin)
```

The `randomWalk` controller encodes each part of the above specification as a separate function, sharing a memory `Mode` that encodes the different robot states. The ROS-style computation graph is illustrated in Fig. 2, where nodes amount to the defined reactive functions, and topics are either user-defined events or those provided for the Kobuki controller. The `bumper` and `cliff` functions change the LED colors, and set in motion a change in direction. For greater modularity, they both emit a new event of type `ChgDir`. The `wheel` function also changes the LEDs' colors and sets the memory mode to the `Stop` state. The `spin` function reads the memory mode, sets the respective velocity depending on the mode, and returns an updated mode. It receives the current time in `Seconds` to determine if the estimated turning time has elapsed.

The most complicated behavior is left to function `chgdir`, that resolves `ChgDir` events to concrete `Turn` actions. To implement random behavior, it resorts to a Haskell standard randomness generator of type `StdGen` to generate a random direction and angle, and reads the current time to calculate the time limit for the `Turn` action. This is a good example of how the power of the full Haskell language can be gradually unleashed as students tackle more advanced problems.

Another popular Kobuki controller is the safety controller³, that imposes stricter conditions on dangerous events. Its C++ ROS implementation can be encoded in ROSY as a single controller that reacts to bumper, cliff or wheel drop events and cautiously decides on a new velocity to escape danger:

Example 6 (Safety controller).

```
safetyControl :: Either (Either Bumper Cliff) Wheel
              -> Maybe Velocity
safetyControl = ...
```

The code for `safetyControl` is conceptually simple, yet verbose as it explores multiple combinations of sensor inputs. We omit it in the paper, but it can be found at the ROSY website.

The safety controller does not do much by itself, and it is typically deployed together with the random walker to limit its actions as the robot roams around. Since both controllers publish possibly conflicting `Velocity` commands to the robot, the traditional ROS solution is to use a multiplexer that remaps topics and allows one controller at a time to command the robot, according to a fixed set of priorities.

We can define a general multiplexer in ROSY as follows:

Example 7 (Binary multiplexer, M1 with priority over M2).

```
data M = Start | Ignore Seconds
data M1 a = M1 a
data M2 a = M2 a

mux :: Seconds -> Memory M
```

²https://github.com/yujinrobot/kobuki/tree/devel/kobuki_random_walker

³https://github.com/yujinrobot/kobuki/tree/devel/kobuki_safety_controller

```

-> Either (M1 a) (M2 a) -> Maybe (a, Memory M)
mux t _ (Left (M1 a)) = Just (a, Memory (Ignore (t+d)))
mux t (Memory (Ignore s)) _ | s > t = Nothing
mux t _ (Right (M2 a)) = Just (a, Memory Start)

```

This binary multiplexer reads the current time t in **Seconds** and reacts to events either marked as **M1** or **M2**, giving higher priority to **M1** events by setting a time interval starting at t and ending at $t+d$ (for a fixed duration d) during which all **M2** events are ignored.

We can then instantiate a safe random walker by remapping output velocities of the safety and random walker controllers with **M1** and **M2** tags and running a multiplexer in parallel:

Example 8 (Random walker with safety controller).

```

safetyControl :: ... -> Maybe (M1 Velocity)
spin :: ... -> (M2 Velocity, ...)
muxVel :: ... -> Either (M1 Velocity) (M2 Velocity)
        -> Maybe (Velocity, ...)

safeRandomWalk = (randomWalk, safetyControl, muxVel)

```

The complete refactored code for the `safeRandomWalk` controller is available at the **ROSY** website.

C. Revisiting block-based languages

Because controllers in **ROSY** run forever, they do not directly lend themselves to performing sequences of instructions. For concreteness, imagine that we want to command the robot to draw a square on the floor with its movement. In a visual robot programming language, this can be done by assembling a block like the one from Fig. 1. Even though such a block can be expressed in **ROSY** as a multi-stage controller that explicitly encodes a state machine and reacts differently depending on the state⁴, it is useful to lend more structure to the language as the complexity of the controller grows.

Like other FRP languages [22], **ROSY** introduces the concept of tasks, as a combination of an initialization step and a continuous controller. The initializer sets up the stage for the controller, that runs continuously until a predefined terminating event occurs. For example, we can make the robot turn sideways by a fixed amount of degrees by writing a task:

Example 9 (Task: Turn left or right).

```

type Side = Either Degrees Degrees

turn :: Side -> Task ()
turn s = task (startTurn s) runTurn

startTurn :: Side -> Orientation
          -> Memory Orientation
startTurn s o = case s of
  Left a -> Memory (o+degreesToOrientation a)
  Right a -> Memory (o+degreesToOrientation a)

runTurn :: Memory Orientation -> Orientation
        -> Either (Velocity) (Done ())
runTurn (Memory to) from = if abs d <= err
  then Right (Done ())
  else Left (Velocity 0 (orientation d))
  where d = normOrientation (to-from)

```

⁴Classical FRP frameworks tackle this general problem by designing advanced higher-order switching combinators over reactive functions, that are expressively powerful but even less novice-friendly.




Fig. 3: The CodeWorld-powered **ROSY** environment.

At initialization, the `startTurn` reads the robot's **Orientation** from its odometry information, and writes the desired final **Orientation** to memory by adding or subtracting the received angle to the current orientation. The `runTurn` controller will rotate the robot towards the desired orientation until the desired and current orientations are equal with a small error margin `err`, signalling when it is **Done**. The **Done** type allow returning additional information on task termination. Returning nothing is achieved with the empty type `()`.

A similar task makes the robot move a fixed distance:

Example 10 (Task: Move forward or backwards).

```

data Direction = Forward Centimeters
              | Backward Centimeters

move :: Direction -> Task ()
move = ...

```

Since tasks can end, in contrast to controllers, we can now mimic the block from Fig. 1 in **ROSY** by using Haskell's monadic notation to sequence tasks in an imperative style:

Example 11 (Task: Draw a square).

```

drawSquare :: Task ()
drawSquare = replicateM_ 4 $ do
  move (Forward 2)
  turn (Left 90)

```

IV. ENVIRONMENT

As well as striving to allow students to learn a “real” programming language while freeing them from inessential technical language details outside of how to control a robot, **ROSY** comes with a fully integrated development environment. To really allow students to concentrate on the meaning of their programs, ignoring deployment details, the **ROSY** environment runs as a web application inside any modern web browser.

Figure 3 shows the **ROSY** environment in action. It is powered by Codeworld [20], a modern educational environment for writing graphical Haskell programs such as games and animations. CodeWorld has been used in K12 schools for years, and supports a code editor with features such as syntax highlighting, improved on-the-fly compiler error messages, extensive documentation or easy sharing of projects.

Another vital component of the environment is a visualizer that simulates, directly in the browser, how the controller




Fig. 4: The ROSY architecture.

programmed by the student interacts with a robot in a fictitious 2D world. At the moment, this is tailored for a TurtleBot2 placed in a tiled world made of floor, walls and cliffs. In ROSY training sessions for K12 students that we have hosted at the University of Minho, students could also deploy their same code to control a real robot and perceive the differences between a simulated and a real world. Enabling students to simultaneously test their examples in simulated and real scenarios played an important role in teaching them the importance of these differences and their influence on the design of approximate, event-driven robotic controllers.

V. UNDER THE HOOD

Despite their simplicity, ROSY programs are fully compatible with the ROS infrastructure. Under the hood, the ROSY language is implemented as an embedded domain-specific language that provides an additional abstraction layer over the existing Haskell ROS library. The ROSY environment is also implemented in Haskell, by extending and specially tailoring the CodeWorld [20] environment to the ROSY language. This includes a custom prelude that is imported by default, a custom code pre-processor to automatically derive necessary Haskell type class instances, and a custom graphical simulation of the robot. All the development source code is open-source and freely available⁵.

A. Haskell ROS library

The ROSHASK library [25] enables ROS programming within the Haskell ecosystem by supporting the deployment of ROS client nodes that are compatible with the rostop communications protocol. ROSHASK fades the architectural boundaries between ROS system and software components, by lifting topics to first-class values and designing a collection of FRP-style combinators to split, fuse and generally manipulate topics in a more expressive, modular and compositional way.

At its core, the library provides functions for subscribing and publishing topics:

```
subscribe :: String -> Node (Topic m a)
publish   :: String -> Topic m a -> Node ()
```

The type variable m is a monad for actions with effects, typically `IO` for interacting with a non-pure outside world, and a is the type of the subscribed or published event derived from standard ROS message type definition files. Topics

are modelled as infinite monadic streams [18], i.e., monadic actions that produce the next value and a new topic:

```
newtype Topic m a = Topic (m (a, Topic m a))
```

The `Node` monad manages TCP connections and internal buffers of published and subscribed messages. The following example node subscribes to two sensors, fuses them using the `bothNew` combinator (subsampling the faster topic), maps an action `act :: (Sense1, Sense2) -> Cmd` over each pair of sensor values, and publishes the resulting commands:

```
n1 = do t1 <- subscribe "sense1"
        t2 <- subscribe "sense2"
        publish "cmd" $ fmap act $ t1 `bothNew` t2
```

Internally, the asynchronous ROSHASK behavior is implemented on top of Haskell user-space threads, that are managed by the Haskell runtime and much more efficient than system threads. For instance, a typical `publish` implementation launches a thread that infinitely samples values from a topic and communicates them to the ROS master; similarly, merging two topics is done by launching two threads that independently consume each topic and write to a common channel.

Nodes can also be easily composed, for instance, we can simultaneously install a handler that listens to and prints the published commands to the command line:

```
n2 = subscribe "cmd" >>= runHandler putStrLn
```

In the style of `publish`, `runHandler` is a general combinator that launches a thread that will consume values from a topic and execute some user-defined IO action:

```
runHandler :: (a -> IO b) -> Topic IO a -> Node ()
```

The composite node `n1 >> n2` will communicate the commands produced by `n1` both via `rostop` and locally to `n2`.

B. Architecture

The ROSY web environment, depicted on left side of Fig. 4, is designed according to the common Model–View–Controller pattern, with components implemented as separate ROSHASK controllers that communicate locally and form a single ROSHASK node. The model controller (loosely) simulates a standard `kobuki_node`⁶ in a simulated world, the view controller implements a simple 2D animation of the robot within the world, and the user controller represents the node specified by the programmer. To support web-based simulation and loosen the dependency on ROS, we have adapted the ROSHASK library to run without a ROS master server. This way, all the Haskell code needed to perform a simulation is compiled via GHCJS into a JavaScript application that runs directly in the client browser.

Alternatively to web-based simulation, ROSY programs can also be executed on a local machine (right side of Fig. 4), by connecting to a ROS master server and operate a more realistic Gazebo simulator or a real Kobuki robot. We have tested both scenarios on Ubuntu 14.04 with ROS Indigo and a TurtleBot2.

⁵<https://github.com/hpacheco/codeworld-rosy>

⁶http://wiki.ros.org/kobuki_node

C. Implicit stream programming

The greatest design decision of the **ROSY** language is that controllers have an implicit notion of time. This favors a simpler declarative style focused on *which* commands are issued *when* events happens, without specifying *how often* subscribers and publishers interact with the **ROS** world, and freeing programmers from common robotic programming details such as clocks, sampling rates or synchronization. Therefore, unlike **ROSHASK**, that exposes a full-fledged API to manipulate topics as a whole, **ROSY** controllers are less expressive in that they only consider a single point in time.

In **ROSY**, events are identified by their type. We define two type classes that internally bind **ROSY** types and **ROS** message namespaces to streams of subscribed sensors or published commands:

```
class Sensor a where
  sensor :: Node (Topic IO a)
class Command a where
  command :: Topic IO a -> Node ()
```

For example, the same **Velocity** type can simultaneously express the act of getting the current velocity from the robot's periodic odometry data and the act of setting the desired velocity by sending a command to the robot's base:

```
instance Sensor Velocity where
  sensor = subscribe "odom"
  >>= return . fmap (_twist . _twist))
instance Command Velocity where
  command = publish
    "/mobile_base/commands/velocity"
```

The **ROSY** language currently supports a fixed set of events offered by the `kobuki_node` API. Extending support for other robots simply requires defining new boilerplate **Sensor** and **Command** instances, as **ROSHASK** already supports many standard **ROS** message types and allows deriving Haskell types from custom **ROS** message files.

To conciliate whole topics with point-wise controllers, we define another type class that implicitly lifts a function on individual values to a controller on streams of values:

```
class Controller a where
  controller :: a -> Node ()
```

The stream semantics of the lifted controller is then inferred from the function's type signature: inputs correspond to subscribed sensors, and outputs to published commands, e.g.:

```
instance (Sensor a, Command b)
  => Controller (a -> b) where
  controller f = sensor >>= command . fmap f
```

Instances for composite types perform implicit stream programming. For example, a **Controller** that receives an input pair is fusing data from two sensors:

```
instance (Sensor a, Sensor b) => Sensor (a,b) where
  sensor = liftM2 bothNew sensor sensor
```

As another example, a **Controller** that returns possibly different commands is splitting the output stream (using the `tee` **ROSHASK** combinator that duplicates a topic), and processing each type of commands independently:

```
instance (Command a, Command b)
  => Command (Either a b) where
  command t = do (t1,t2) <- tee t
                 command $ lefts t1
                 command $ rights t2
```

Multiple **Controllers** are executed in parallel threads, and can be composed in sequence:

```
instance (Controller a, Controller b)
  => Controller (a,b) where
  controller (a,b) = controller a >> controller b
```

D. Supporting user-defined events and memory

ROSY also allows the declaration of user-defined data types for more modular intra-node communication. Since these need not be bound to **ROS** namespaces, every newly declared user-defined data type is by default a **Sensor** and a **Command**. This is supported by a new **UserNode** monad that extends **Node** capabilities with local type-indexed event buffers for user-defined data types.

Since **ROSY** programs are by design pure Haskell functions, there is no native support for common robotics design patterns that global memory. It is nonetheless possible to emulate global variables by publishing an initial event, and on every update subscribing to current event and re-publishing its updated value. Even so, this pattern can be error-prone as the programmer needs to be cautious about subscribing and publishing the “variable” the right number of times in order to keep it alive. This may also be problematic if more than one controller is manipulating the same “variable” in parallel.

To avoid these caveats, **ROSY** supports node-specific memory, distinguishable through a type-level tag:

```
data Memory a = Memory a
```

We implement memory by extending the **UserNode** monad with a transactional memory store, holding a global value for every distinct type `a`.

In order to support transactional controllers, or more specifically, be able to execute each controller thread as a single transaction, we must generalize our sensor and command interfaces to produce and consume topics of transactions⁷:

```
sensor :: UserNode (Topic IO (STM a))
command :: Topic IO (STM (Maybe a))
         -> UserNode (Topic IO (STM ()))
```

A **Sensor** (`Memory a`) returns a topic that repeatedly reads from the memory variable of type `a`, while a **Command** (`Memory a`) appends memory writes to a topic of transactions, returning a new topic. The greatest change occurs on commands that, unlike before, must be published synchronously within the same transaction, meaning we can no longer fork a topic and publish each side independently. We can execute a controller thread by installing a handler that `atomically` executes each transaction as a side effect:

```
instance (...) => Controller (a -> b) where
  controller f = sensor >>= command . fmap f
  >>= lift . runHandler atomically
```

E. Sequencing tasks

In **ROSHASK** and other FRP approaches, topics are modelled as infinite streams and topic handlers are program-long threads continuously waiting on and reacting to events. The fact that the data flow graph, inferred from the wiring of stream

⁷The **STM** monad stands for Haskell's software transactional memory library. The **Maybe** type is a technical requirement for filtering values inside a transaction, since instances must not change the periodicity of the topics.

combinators, is typically known statically, allows **ROSHASK** to register all subscribers and publishers with the **ROS** master at node initialization, before starting to actually process data.

In **ROSY**, a task is defined as an initialization action, a continuous controller, and a terminating event:

```
task :: (Command init, Controller ctrl)
  => init -> ctrl -> Task end
```

A controller issues termination via a special event type:

```
data Done a = Done a
```

We also make **Task** a monad, so that programmers can use monadic notation to sequence tasks. For composing two smaller tasks into a composite task, where the output of the first is passed on to the second, we may write:

```
task12 = do end1 <- task init1 ctrl1
         task (init2 end) ctrl2
```

In this scenario, controllers no longer run forever: when the first task ends, we must uninstall the controller `ctrl1` and install a new controller `ctrl2` for the second task. We have extended **ROSHASK** to support dynamic node configuration: each task runs within its own **UserNode**; publishers and subscribers are registered at declaration time; tasks keep a fine-grained control of launched threads, and all children threads are killed when exiting the parent **UserNode**⁸.

VI. CONCLUSION

In this paper we have presented **ROSY**, a new pedagogical robot programming language that advocates a sweet-spot between the expressiveness of FRP and the needed simplicity of an educational setting. As part of a computing summer camp for held at the University of Minho⁹, in July 2019, we have taught a 4-hour training session for K12 students on hands-on robot programming in **ROSY**, with only two prior sessions of general programming in Haskell. From our perceptions, students were able to comprehend the concepts and quickly start programming the robot to perform simple tasks.

In the future, we plan to provide further **ROSY** training sessions and undergo empirical studies that can corroborate its practical value for learning programming via robotics. We plan to improve the **ROSY** environment with more advanced simulation scenarios using, e.g., the Gazebo web client¹⁰ or remote **ROS** support similar to [10]. We also plan to explore the design of novel novice-friendly interfaces that blend textual and visual representations, and graphically combine visual blocks typical of imperative robotic approaches with graphical data flow diagrams typical of advanced FRP approaches.

REFERENCES

- [1] D. S. Blank, “Robots make computer science personal,” *Commun. ACM*, vol. 49, no. 12, pp. 25–27, 2006.
- [2] A. Oddie, P. Hazlewood, S. Blakeway, and A. Whitfield, “Introductory problem solving and programming: Robotics versus traditional approaches,” *Innovation in Teaching and Learning in Information and Computer Sciences*, vol. 9, no. 2, pp. 1–11, 2010.

⁸Note that messages are not lost when transitioning between tasks, since a **Node** keeps global buffers of published and subscribed **ROS** topics.

⁹<https://www.uminho.pt/veraonocampus>

¹⁰<http://gazebosim.org/gzweb.html>

- [3] A. Druin, J. A. Hendler, and J. Hendler, *Robots for kids: Exploring new technologies for learning*. Morgan Kaufmann, 2000.
- [4] E. A. Gandy, S. Bradley, D. Arnold-Brookes, and N. R. Allen, “The use of LEGO Mindstorms NXT robots in the teaching of introductory Java programming to undergraduate students,” *Innovation in Teaching and Learning in Information and Computer Sciences*, vol. 9, no. 1, pp. 2–9, 2010.
- [5] P. B. Lawhead, M. E. Duncan, C. G. Bland, M. Goldweber, M. Schep, D. J. Barnes, and R. G. Hollingsworth, “A road map for teaching introductory programming using LEGO Mindstorms robots,” *SIGCSE Bulletin*, vol. 35, no. 2, pp. 191–201, 2003.
- [6] A. Cleary, L. Vandenbergh, and J. Peterson, “Reactive game engine programming for STEM outreach,” in *SIGCSE*. ACM, 2015, pp. 628–632.
- [7] M. Felleisen, R. B. Fidler, M. Flatt, and S. Krishnamurthi, “The TeachScheme! project: Computing and programming for every student,” *Computer Science Education*, vol. 14, no. 1, pp. 55–77, 2004.
- [8] M. Quigley, K. Conley, B. P. Gerkey, J. Faust, T. Foote, J. Leibs, R. Wheeler, and A. Y. Ng, “ROS: An open-source Robot Operating System,” in *ICRA Workshop on Open Source Software*, 2009.
- [9] J. P. Diprose, B. A. MacDonald, and J. G. Hosking, “Ruru: A spatial and interactive visual programming language for novice robot programming,” in *VL/HCC*. IEEE, 2011, pp. 25–32.
- [10] G. A. Casañ, E. Cervera, A. A. Moughlby, J. Alemany, and P. Martinet, “ROS-based online robot programming for remote education and training,” in *ICRA*. IEEE, 2015, pp. 6101–6106.
- [11] I. Angulo, J. García-Zubía, U. Hernández-Jayo, I. Uriarte, L. Rodríguez-Gil, P. Orduña, and G. M. Pieper, “RoboBlock: A remote lab for robotics and visual programming,” in *exp.at*. IEEE, 2017, pp. 109–110.
- [12] M. H. Masum, T. S. Rifat, S. M. Tareeq, and H. Heickal, “A framework for developing graphically programmable low-cost robotics kit for classroom education,” in *ICETC*. ACM, 2018, pp. 22–26.
- [13] D. Weintrop, A. Afzal, J. Salac, P. Francis, B. Li, D. C. Shepherd, and D. Franklin, “Evaluating CoBlox: A comparative study of robotics programming environments for adult novices,” in *CHI*. ACM, 2018, pp. 366:1–366:12.
- [14] D. Marghitu and S. Coy, “Robotics rule-based formalism to specify behaviors in a visual programming environment,” in *B&B@VL/HCC*. IEEE, 2015, pp. 45–47.
- [15] D. Weintrop, “Block-based programming in computer science education,” *Commun. ACM*, vol. 62, no. 8, pp. 22–25, 2019.
- [16] Z. Xu, A. D. Ritzhaupt, F. Tian, and K. Umaphathy, “Block-based versus text-based programming environments on novice student learning outcomes: A meta-analysis study,” *Computer Science Education*, vol. 29, no. 2-3, pp. 177–204, 2019.
- [17] E. Bainomugisha, A. L. Carreton, T. V. Cutsem, S. Mostinckx, and W. D. Meuter, “A survey on reactive programming,” *ACM Comput. Surv.*, vol. 45, no. 4, pp. 52:1–52:34, 2013.
- [18] I. Perez, M. Bärenz, and H. Nilsson, “Functional reactive programming, refactored,” in *Haskell*. ACM, 2016, pp. 33–44.
- [19] M. Felleisen, R. B. Fidler, M. Flatt, and S. Krishnamurthi, “A functional I/O system or, fun for freshman kids,” in *ICFP*. ACM, 2009, pp. 47–58.
- [20] J. Breitter and C. Smith, “Lock-step simulation is child’s play (experience report),” *PACMPL*, vol. 1, no. ICFP, pp. 3:1–3:15, 2017.
- [21] J. B. Almeida, A. Cunha, N. Macedo, H. Pacheco, and J. Proença, “Teaching how to program using automated assessment and functional Glossy games (experience report),” *PACMPL*, vol. 2, no. ICFP, pp. 82:1–82:17, 2018.
- [22] J. Peterson, P. Hudak, and C. Elliott, “Lambda in motion: Controlling robots with haskell,” in *PADL*, ser. LNCS, vol. 1551. Springer, 1999, pp. 91–105.
- [23] P. Hudak, A. Courtney, H. Nilsson, and J. Peterson, “Arrows, robots, and functional reactive programming,” in *Advanced Functional Programming*, ser. LNCS, vol. 2638. Springer, 2002, pp. 159–187.
- [24] I. Pembeci, H. Nilsson, and G. D. Hager, “Functional reactive robotics: An exercise in principled integration of domain-specific languages,” in *PPDP*. ACM, 2002, pp. 168–179.
- [25] A. Cowley and C. J. Taylor, “Stream-oriented robotics programming: The design of roshask,” in *IROS*. IEEE, 2011, pp. 1048–1054.